Multilevel Dynamic Twin Modeling

N.K. Schuurman

Utrecht University

Y. Zheng

University of Alberta

C.V. Dolan

VU Amsterdam

Author Note

The authors are grateful to Marieke Wichers, Nele Jacobs, Jim van Os, Catherine Derom, and Evert Thiery for sharing the data for the empirical example. The authors are grateful to to Maria Bolsinova, Ellen Hamaker and Karianne Schuurman for insightful conversations on this work.

This work was partly supported by the Natural Sciences and Engineering Research Council of Canada (Discovery Grant RGPIN-2020-04458 and Discovery Launch Supplement DGECR-2020-00077 to Yao Zheng).

Corresponding author address: Methodology & Statistics, Padualaan 14, 3584 CH Utrecht, The Netherlands. e-mail address: mail@nkschuurman.com

Multilevel Dynamic Twin Modeling

Supplemental: Specification of correlated random means μ_{ij} in Mplus

In the multilevel model we assume that the trait scores μ_{ij} are multivariate normally distributed (twin member 1 is jointly distributed with twin member 2), where MZ twin member 1 and MZ twin member 2 are modeled with the same mean $\gamma_{\mu MZ}$ and variance $\psi^2_{\mu MZ}$, and a covariance $\psi_{\mu MZ1MZ2}$, and DZ twin member 1 and DZ twin member 2 are modeled with the same mean $\gamma_{\mu DZ}$ and variance $\psi^2_{\mu DZ}$, and a covariance $\psi_{\mu DZ1DZ2}$. In Mplus DSEM v8 the option of specifying a level 2 covariance matrix with variables that have the same variance (μ_{MZ1j} and μ_{MZ2j}, and μ_{DZ1j} and μ_{DZ2j}), but have different covariances is not provided. Given that there is no particular reason to assume that the group level variances and means of two members of the same twins are different, we specified the above variance-covariance structure by introducing latent variables.

Specifically, we specify for the MZ twins: $\mu_{1j} = C_j + \omega_{1j}$ and $\mu_{2j} = C_j + \omega_{2j}$. Here C_j is a shared latent variable for both members of the same twin, and its variance is equal to the covariance $\psi_{\mu MZ1MZ2}$ of the means of MZ members 1 and MZ members 2; note that this covariance is hence restricted to be positive. The residuals ω_{ij} capture the part of the random means that is unique to each twin member. The residual variances of ω_{1j} for twin members 1 and ω_{2j} for twin members 2 are restricted to be equal. Hence, we get the following total variance for the random means of MZ twins: $\psi^2_{\mu MZ} = var(C_j) + var(\omega_j)$. The same structure is implemented for the DZ twins.

Supplemental: Mplus model code

```
DATA: FILE = Mplus_MZDZWich_zen.dat;
ANALYSIS:  TYPE IS TWOLEVEL RANDOM;
 ESTIMATOR=bayes; FBITER=(200000); PROC=2;BSEED=5 6;
SAVEDATA: bparameters=bp_szen_ziclv_mulat3_2v.dat;
    save = fs(1000); file=fs_szen_ziclv_mulat3_2v.dat;
```

```
     factors= all;
     data imputation: thin=100;
VARIABLE:NAMES = clus mzen1 mzen2 dzen1 dzen2;
MISSING ARE *;
CLUSTER = clus;
model:
   %WITHIN%
      fedz1| Edz1 by dzen1(&1); !random loadings for dz twin 1 on E1_dz
      fedz2| Edz2 by dzen2(&1); !random loadings for dz twin 2 on E2_dz
      fadz1| Adz1 by dzen1(&1); !random loadings for dz twin 1 on A1_dz
  fadz2| Adz2 by dzen2(&1);   !random loadings for dz twin 2 on A2_dz
      femz1| Emz1 by mzen1(&1); !random loadings for mz twin 1 on E1_mz
      femz2| Emz2 by mzen2(&1); !random loadings for mz twin 2 on E2_mz
      famz1 famz2| Amz by mzen1  mzen2(&1); !random loadings for mz twin 1 and 2 on A
           !(the same A factor for both MZ twins)

       phiedz1| Edz1 on Edz1&1; !random autoregressive effect for dz twin 1 for E1_dz
       phiedz2| Edz2 on Edz2&1; !random autoregressive effect for dz twin 2 for E2_dz
       phiadz1 | Adz1 on Adz1&1; !random autoregressive effect for dz twin 1 for A1_dz
    phiadz2 | Adz2 on Adz2&1; !random autoregressive effect for dz twin 2 for A2_dz
    phiemz1| Emz1 on Emz1&1;  !random autoregressive effect for mz twin 1 for E1_mz
       phiemz2| Emz2 on Emz2&1; !random autoregressive effect for mz twin 2 for E2_mz
       phiamz | Amz on Amz&1; !random autoregressive effect, the same for both mz twins

  !scaling in the variances of the factors:
       Edz1@1;
       Edz2@1;
```

```
        Adz1@1;
        Adz2@1;
        Emz1@1;
        Emz2@1;
        Amz@1;

!Random logtransformed residual variances for the measurement model of each twin:
        logvdzen1 |dzen1;
        logvdzen2 |dzen2;
        logvmzen1 |mzen1;
        logvmzen2 |mzen2;

!Correlations among the different A and E factors fixed to zero:
        Edz1 with Edz2@0;
        Edz1 with Adz1@0 ;
        Edz2 with Adz2@0 ;
        Edz1 with Adz2@0 ;
        Edz2 with Adz1@0 ;

!specify a latent variable on which both Adz factor load, random effect here results
!in random correlations between the A factors of the dz twins
        Dzcov by  Adz1@1 Adz2@1;
        cordz | Dzcov;

        !Correlations among the different A and E factors fixed to zero:
        Emz1 with Emz2@0;
        Emz1 with Amz@0 ;
```

```
        Emz2 with Amz@0 ;

     Emz1 with Edz1@0;
     Emz1 with Edz2@0;
     Emz1 with Adz1@0;
     Emz1 with Adz2@0;

        Emz2 with Edz1@0;
     Emz2 with Edz2@0;
     Emz2 with Adz1@0;
     Emz2 with Adz2@0;

        Amz with Edz1@0;
     Amz with Edz2@0;
     Amz with Adz1@0;
     Amz with Adz2@0;

  %BETWEEN%
    [dzen1*.55](1); !average mean phenotype for dz1
    dzen1*0.3 (20); !variance mean phenotype for dz1
    [dzen2*.55](1); !average mean phenotype for dz2 (same as dz1)
    dzen2*0.3 (20); !variance mean phenotype for dz2 (same as dz1)
   [mzen1*.55](1);  !average mean phenotype for mz1
   mzen1*0.3 (30);  !variance mean phenotype for mz1
   [mzen2*.55](1);  !average mean phenotype for mz2 (same as mz1)
   mzen2*0.3 (30);  !variance mean phenotype for mz2 (same as mz1)
```

```
   dzmucor by  dzen1@1 dzen2@1; !allow means of dz twins to correlate via latent variabl
    mzmucor by  mzen1@1 mzen2@1; !allow means of mz twins to correlate via latent variab
    dzmucor with mzmucor@0;

    [cordz@.5]; !mean correlation among A factors of the dz twins set to 0.5
   cordz*0.01; !variance correlation among A factors of the dz twins

!means and variances for the random autoregressive coefficients
   [phiedz1*.3] (3);
    phiedz1*.01 (4);
      [phiedz2*.3](3);
    phiedz2*.01 (4);
    [phiemz1*.3] (3);
    phiemz1*.01 (4);
      [phiemz2*.3](3);
    phiemz2*.01 (4);

     [phiadz1*.7](5);
    phiadz1*.005(6);
   [phiadz2*.7](5);
    phiadz2*.005(6);
    [phiamz*.7](5);
    phiamz*.005(6);

!means and variances for the random loadings
     [fadz1*.55](7);
     [fadz2*.55](7);
```

```
    fadz1*.008(8);
    fadz2*.008(8);
   [famz1*.55](7);
     [famz2*.55](7);
    famz1*.008(8);
    famz2*.008(8);

   [fedz1*.63](9);
     [fedz2*.63](9);
    fedz1*.01(10);
    fedz2*.01(10);
  [femz1*.63](9);
     [femz2*.63](9);
    femz1*.01(10);
    femz2*.01(10);

!means and variances for the random logtransformed residual variances
  [logvdzen1*-.6] (12);
   [logvdzen2*-.6] (12);
   [logvmzen1*-.6] (12);
   [logvmzen2*-.6] (12);

   logvdzen1*.1 (11);
   logvdzen2*.1 (11);
   logvmzen1*.1 (11);
   logvmzen2*.1 (11);
```

```
PLOT:
  TYPE IS PLOT1 PLOT2 PLOT3;
  OUTPUT:TECH1 TECH8;
```